*-------------------------------------------------------------------------------*;

*- %glm2x2:  Test Assumptions, Run model, summarize results of a 2x2 2-way GLM -*;

*-------------------------------------------------------------------------------*;

/*

Macro %GLM2X2 read_me notes:

Inputs:

indata


Name of the input dataset

depvar


Name of the dependent variable to be modelled

factor1


Name of the first 2-level factor (independent variable)

factor2


Name of the second 2-level factor (independent variable)

deplabel

Label of the dependent variable to appear in the printed results summary

outdata


SAS dataset name to output the results summary into

print


Option to suppress automatic printed output. 





Set to "Y" for "yes" as default, to print the output.





Set to anything besides Y to surpress printing.





Not required to input.

Returns:

Prints (to SAS Listing) a dataset that summarizes the output of a 2x2 2-way GLM model including: number of observations,

Normality and Equal Variance tests, ANOVA table, Least Squares Means and All pairwise comparisons with multiple adjusted

p-values reported. The standard printing can be suppressed while the dataset remains available in the WORK library under

the name assigned in the macro call. 

Example calls:

--This example requests the standard printed output:

%glm2x2(indata=all0r, depvar=bvtv_femur, factor1=geno, factor2=treat, deplabel=BVTV Femur, outdata=tf0_none, print=Y);

--This example suppresses the standard printed output:

%glm2x2(indata=all0r, depvar=bvtv_femur, factor1=geno, factor2=treat, deplabel=BVTV Femur, outdata=tf0_none, print=N);

--This example prints the standard output by default:

%glm2x2(indata=all0r, depvar=bvtv_femur, factor1=geno, factor2=treat, deplabel=BVTV Femur, outdata=tf0_none);

Other notes:

--Also uses 8 Dummy Datasets (_title1-_title8) that are defined in the program this program, but outside of the macro itself.

  If the datasets are deleted, a macro call will not run correctly. This entire program needs to be re-run.

  These dummy datasets merely provide labelling and spacing for the output dataset/table.

--The macro assumes that when ascendingly sorted the 1st level of each factor is "control" and second is "experimental". 

  Though this distinction may only matter for automatic choices of the 4 comparisons adjusted for by the Holm & Hochberg procedures.

*/

*-------------------------------------------------------------------------------*;

*-------------------------- "Dummy" Datasets - START ---------------------------*;

/**Dummy Datasets defined outside of the macro since "cards" statements cannot be used inside a SAS macro**/

**--Dummy Datasets for header and blank lines in output table dataset--**;

data _title1;

length col1 $ 30 col2 $ 30 col3 $ 30;

input col0 $ col1 $ col2 $ col3 $;

cards;

Blank Blank Blank Blank

Blank Test Read Used

;

run;

data _title2;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30 col5 $ 30 col6 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $ col5 $ col6 $;

cards;

Blank Blank Blank Blank Blank Blank Blank

Blank Normality-Tests: p(Overall) p(1) p(2) p(3) p(4)

;

run;

data _title3;

length col1 $ 30 col2 $ 30 col3 $ 30;

input col0 $ col1 $ col2 $ col3 $;

cards;

Blank Blank Blank Blank

;

run;

data _title4;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30 col5 $ 30 col6 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $ col5 $ col6 $;

cards;

Blank Blank Blank Blank Blank Blank Blank

Blank ANOVA Table Blank Blank Blank Blank

Blank Source DF SS MS F P

;

run;

data _title5;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $;

cards;

Blank Blank Blank Blank Blank

Blank Least Squares Means Blank

Blank Genotype Diet LSmean p-value

;

run;

data _title6;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $;

cards;

Blank Blank Blank Blank

;

run;

data _title7;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $;

cards;

Blank Blank Blank Blank

;

run;

data _title8;

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30 col5 $ 30 col6 $ 30 col7 $ 30;

input col0 $ col1 $ col2 $ col3 $ col4 $ col5 $ col6 $ col7 $;

cards;

Blank Blank  Blank    Blank         Blank    Blank    Blank   Blank

Blank All    Pairwise Comparisons   Blank    Blank    Blank   Blank

Blank GroupA GroupB   Unadjusted(p) Tukey(p) Sidak(p) Holm(p) Hochberg(p)

;

run;

*-------------------------- "Dummy" Datasets --- END ---------------------------*;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*--------------------------- GLM 2x2 macro -- START ----------------------------*;

%macro glm2x2(indata=, depvar=, factor1=, factor2=, deplabel=, outdata=, print=Y);

*create main temporary dataset from input dataset;

data _main_temp;

set &indata;

*create interaction macro variable from values of factor1 & factor2 macro variables;

call symput ("interaction", "&factor1"||'_'||"&factor2");

run;

run; ods listing close; run;

*create 4-level combined factor variable;

proc freq data=_main_temp;

tables &factor1*&factor2 / list;

ods output List=_list;

run;

data _list(keep= &factor1 &factor2 _4level_factor_n _4level_factor_c);

set _list;

_4level_factor_n=_n_;

_4level_factor_c=compress(_n_,' ');

*create macro variable labels for the 4-groups of the 2 crossed factors;

if _4level_factor_n=1 then call symput ("label_group1", "1)"||compress(&factor1,' ')||'/'||compress(&factor2,' '));

 else if _4level_factor_n=2 then call symput ("label_group2", "2)"||compress(&factor1,' ')||'/'||compress(&factor2,' '));

 else if _4level_factor_n=3 then call symput ("label_group3", "3)"||compress(&factor1,' ')||'/'||compress(&factor2,' '));

 else if _4level_factor_n=4 then call symput ("label_group4", "4)"||compress(&factor1,' ')||'/'||compress(&factor2,' '));

run;

proc sort data=_list; by &factor1 &factor2;

proc sort data=_main_temp; by &factor1 &factor2;

data _main_temp;

merge _main_temp _list;

by &factor1 &factor2;

run;

****----Procs with ODS-suppressed output----****;

**--1 combined factor GLM for Equal Variance test--**;

proc glm data=_main_temp;

class _4level_factor_c;

model &depvar=_4level_factor_c;

means _4level_factor_c / hovtest;

ods output HOVFTest=_hov;

run;

**--1 combined factor GLM for Equal Variance test--**;

**--Main GLM--**;

proc glm data=_main_temp;

class &factor1 &factor2;

model &depvar=&factor1 &factor2 &factor1*&factor2 / solution;

lsmeans &factor1 / pdiff;

lsmeans &factor1 / adjust=tukey;

lsmeans &factor1 / adjust=sidak;

lsmeans &factor2 / pdiff;

lsmeans &factor2 / adjust=tukey;

lsmeans &factor2 / adjust=sidak;

lsmeans &factor1*&factor2 / pdiff;

lsmeans &factor1*&factor2 / adjust=tukey;

lsmeans &factor1*&factor2 / adjust=sidak;

output out=_resid2 r=resid;

ods output Nobs=_nobs OverallANOVA=_oa ModelANOVA=_ma LSMeans=_lsm Diff=_diff;;

run;

**--Main GLM--**;

**--Univariate for Normality test--**;

proc univariate data=_resid2 normal;

var &depvar resid;

ods output TestsForNormality=_norm;

run;

**--Univariate for Normality test--**;

**--Normality test by groups--**;

proc sort data=_resid2; by &factor1 &factor2;

proc univariate data=_resid2 normal;

by &factor1 &factor2;

var &depvar resid;

ods output TestsForNormality=_normby;

run;

**--Normality test by groups--**;

run; ods listing; run;

****----Procs with ODS-suppressed output----****;

****----Organizing output datasets----****;

**--Normality--**;

data _norm(keep= varname pvalue);

set _norm;

if test="Shapiro-Wilk"; *Other normal test options: "Shapiro-Wilk","Kolmogorov-Smirnov","Cramer-von Mises","Anderson-Darling";

run;

**--Normality--**;

**--Normality by group--**;

proc sort data=_normby; by &factor1 &factor2;

proc sort data=_list; by &factor1 &factor2;

data _normby(keep= varname pvalue &factor1 &factor2 _4level_factor_n _4level_factor_c);

merge _normby _list;

by &factor1 &factor2;

if test="Shapiro-Wilk";

run;

proc sort data=_normby; by varname _4level_factor_n;

proc transpose data=_normby out=_normby2 prefix=p_;

by varname;

id _4level_factor_n;

var pvalue;

run;

proc sort data=_norm; by varname;

proc sort data=_normby2; by varname;

data _norm2(keep= VarName p_overall p_1 p_2 p_3 p_4 passed);

merge _norm _normby2;

by varname;

length passed $ 10;

if .<min(pvalue,p_1,p_2,p_3,p_4)<=0.05 then passed="Failed";

 else if 0.05<min(pvalue,p_1,p_2,p_3,p_4) then passed="Passed";

rename pvalue=p_overall;

run;

**--Normality by group--**;

**--Equality of Variance--**;

data _hov(keep= dependent method ProbF passed);

set _hov;

length passed $ 10;

if .<probf;

if .<probf<=0.05 then passed="Failed";

 else if 0.05<probf then passed="Passed";

run;

**--Equality of Variance--**;

**--# Obs--**;

data _nobs(keep= label nvalue1 DependentVariables);

set _nobs;

run;

proc sort data=_nobs; by dependentvariables;

proc transpose data=_nobs out=_tnobs;

by dependentvariables;

id label;

var nvalue1;

run;

data _tnobs(drop= _name_);

set _tnobs;

rename Number_of_Observations_Read=n_obs_read Number_of_Observations_Used=n_obs_used;

run;

**--# Obs--**;

**--ANOVA Table--**;

data _anova(drop= hypothesistype);

set _ma _oa;

if HypothesisType=3 then delete;

if source="Model" then delete;

run;

**--ANOVA Table--**;

**--Least Square Means--**;

data _lsm_f1 _lsm_f2 _lsm_ia;

set _lsm;

if effect="&factor1" then output _lsm_f1;

 else if effect="&factor2" then output _lsm_f2;

  else if effect="&interaction" then output _lsm_ia;

run;

data _lsm_f1(drop= &factor2 lsmeannumber);

set _lsm_f1;

run;

data _lsm_f2(drop= &factor1 lsmeannumber);

set _lsm_f2;

run;

data _lsm_ia(drop= probtdiff);

set _lsm_ia;

run;

proc sort data=_lsm_f1 nodupkey; by effect dependent &factor1;

proc sort data=_lsm_f2 nodupkey; by effect dependent &factor2;

proc sort data=_lsm_ia nodupkey; by effect dependent &factor1 &factor2;

run;

**--Least Square Means--**;

**--Multiple Comparison P-values--**;

data _diff_f1 _diff_f2 _diff_ia;

set _diff;

if effect="&factor1" then output _diff_f1;

 else if effect="&factor2" then output _diff_f2;

  else if effect="&interaction" then output _diff_ia;

run;

data _diff_ia;

set _diff_ia;

length adjust $ 15;

if 1<=_n_<=4 then adjust="unadjusted";

 else if 5<=_n_<=8 then adjust="tukey";

  else if 9<=_n_<=12 then adjust="sidak";

run;

proc sort data=_diff_ia; by effect dependent adjust rowname;

proc transpose data=_diff_ia out=_tdiff_ia;

by effect dependent adjust rowname;

var _1 _2 _3 _4;

run;

data _tdiff_ia(drop= _name_ effect rowname _label_);*col1;

set _tdiff_ia;

length pvalue $ 12;

if .<col1<=0.05 then pvalue=compress(put(col1,pvalue6.4)," ")||"***";

 else if 0.05<col1 then pvalue=compress(put(col1,pvalue6.4)," ");

groupa=rowname+0;

groupb=_label_+0;

if .<groupa<groupb;

run;

*Calculate Holm-adjusted p-value on only the 4 comparisons typically of interest;

data _holm(drop= pvalue);

set _tdiff_ia;

if adjust='unadjusted';

if (groupa=1 & groupb=2)|(groupa=1 & groupb=3)|(groupa=2 & groupb=4)|(groupa=3 & groupb=4);

rename col1=raw_p;

run;

*Proc MULTTEST to do the adjustments;

run; ods listing close; run;

proc multtest inpvalues=_holm holm hoc;

ods output pValues=_adjp;

run; ods listing; run;

*Rename vars in multtest output data, merge back onto Holm dataset with group numbers;

data _adjp(drop= test);

set _adjp;

rename raw=raw_p StepdownBonferroni=p_holm hochberg=p_hochberg;

run;

proc sort data=_holm; by raw_p;

proc sort data=_adjp; by raw_p;

data _holm(drop= adjust raw_p);

merge _holm _adjp;

by raw_p;

proc sort data=_holm; by groupa groupb;

run;

proc sort data=_tdiff_ia; by dependent groupa groupb;

proc transpose data=_tdiff_ia out=_ttdiff_ia;

by  dependent groupa groupb;

id adjust;

var pvalue;

run;

data _ttdiff_ia(drop= _name_);

set _ttdiff_ia;

run;

*Merge Holm/Hochberg onto main p-value dataset;

proc sort data=_ttdiff_ia; by dependent groupa groupb;

proc sort data=_holm; by dependent groupa groupb;

data _ttdiff_ia(drop= p_holm p_hochberg);

merge _ttdiff_ia _holm;

by dependent groupa groupb;

length holm $ 12 hochberg $ 12;

if .<p_holm<=0.05 then holm=compress(put(p_holm,pvalue6.4)," ")||"***";

 else if 0.05<p_holm then holm=compress(put(p_holm,pvalue6.4)," ");

if .<p_hochberg<=0.05 then hochberg=compress(put(p_hochberg,pvalue6.4)," ")||"***";

 else if 0.05<p_hochberg then hochberg=compress(put(p_hochberg,pvalue6.4)," ");

run;

**--Multiple Comparison P-values--**;

****----Organizing output datasets----****;

**--Output table dataset--**;

data &outdata(keep= order col1-col7);

length col1 $ 30 col2 $ 30 col3 $ 30 col4 $ 30 col5 $ 30 col6 $ 30 col7 $ 30;

set _tnobs(in=in1) _title2 _norm2(in=in2) _title3 _hov(in=in3) _title4 _anova(in=in4)

  _title5 _lsm_f1(in=in5) _title6 _lsm_f2(in=in6) _title7 _lsm_ia(in=in7) _title8 _ttdiff_ia(in=in8);

a=in1; b=in2; c=in3; d=in4; e=in5; f=in6; g=in7; h=in8;

order=_n_;

if col1="Blank" then col1=" "; if col2="Blank" then col2=" "; if col3="Blank" then col3=" ";

if col4="Blank" then col4=" "; if col5="Blank" then col5=" "; if col6="Blank" then col6=" ";

if col7="Blank" then col7=" ";

**;

if in1 then do;

 col1="# of observations";

 col2="Read:";

 col3=compress(n_obs_read," ");

 col4="Used:";

 col5=compress(n_obs_used," ");

end;

if in2 then do;

 if varname="resid" then col1="Residuals";

  else col1=varname;

 col2=compress(put(p_overall,pvalue6.4)," ");

 col3=compress(put(p_1,pvalue6.4)," ");

 col4=compress(put(p_2,pvalue6.4)," ");

 col5=compress(put(p_3,pvalue6.4)," ");

 col6=compress(put(p_4,pvalue6.4)," ");

 col7=passed;

end;

**;

if in3 then do;

 col1="Equal Var. Test";

 col2="p-value=";

 col3=compress(put(probf,pvalue6.4)," ");

 col4=passed;

end;

**;

if in4 then do;

 col1=source;

 col2=compress(df," ");

 col3=compress(put(ss+0,8.4)," ");

 col4=compress(put(ms+0,8.4)," ");

 col5=compress(put(fvalue+0,8.4)," ");

 if source in('Error','Corrected Total') then col6='-';

  else if .<probf<=0.05 then col6=compress(put(probf,pvalue6.4)," ")||"***";

  else if 0.05<probf then col6=compress(put(probf,pvalue6.4)," ");

end;

**;

if in5 then do;

 col1=&factor1;

 col3=compress(put(LSMean,8.4)," ");

 if .<probtdiff<=0.05 then col4=compress(put(probtdiff,pvalue6.4)," ")||"***";

  else if 0.05<probtdiff then col4=compress(put(probtdiff,pvalue6.4)," ");

end;

**;

if in6 then do;

 col2=&factor2;

 col3=compress(put(LSMean,8.4)," ");

 if .<probtdiff<=0.05 then col4=compress(put(probtdiff,pvalue6.4)," ")||"***";

  else if 0.05<probtdiff then col4=compress(put(probtdiff,pvalue6.4)," ");

end;

**;

if in7 then do;

 col1=&factor1;

 col2=&factor2;

 col3=compress(put(LSMean,8.4)," ");

end;

**;

if in8 then do;

 if groupa=1 then col1="&label_group1";

  else if groupa=2 then col1="&label_group2";

  else if groupa=3 then col1="&label_group3";

  else if groupa=4 then col1="&label_group4";

 if groupb=1 then col2="&label_group1";

  else if groupb=2 then col2="&label_group2";

  else if groupb=3 then col2="&label_group3";

  else if groupb=4 then col2="&label_group4";

 col3=unadjusted;

 col4=tukey;

 col5=sidak;

 col6=holm;

 col7=hochberg;

end;

label col1="2-way GLM:" col2="&deplabel" col3="= &factor1" col4="+" col5="&factor2" col6="+" col7="Interaction";

run;

**--Output table dataset--**;

*------Optional printed output------*;

%if &print=Y %then %do;

proc sort data=&outdata; by order;

proc print data=&outdata noobs l;

var col1-col7;

title "2x2 GLM: &depvar = &factor1 + &factor2 + &interaction"; run; title;

run;

%end;

*------Optional printed output------*;

*-------------------------------------------------------------------------------*;

*Delete temporary working datasets in macro;

proc datasets library=work nolist nowarn;

delete _main_temp _list _resid _hov _nobs _oa _ma _lsm _diff _resid2 _norm _normby _normby2 _norm2 _tnobs _anova

 _lsm_f1 _lsm_f2 _lsm_ia _diff_f1 _diff_f2 _diff_ia _tdiff_ia _ttdiff_ia;

run;

quit;

*-------------------------------------------------------------------------------*;

%mend;

*--------------------------- GLM 2x2 macro ---- END ----------------------------*;

*-------------------------------------------------------------------------------*;

run;
